

# Operational Semantics of DistAlgo

Scott D. Stoller

August 18, 2014

This document gives an abstract syntax and operational semantics for a core language for DistAlgo [LSLG12]. The operational semantics is a reduction semantics with evaluation contexts [WF94, SRM09].

## 1 Abstract Syntax

The abstract syntax is defined in Figures 1 and 2. We use some syntactic sugar in sample code. Specifically, (1) we use infix notation for some binary operators, such as **and** and **is**; (2) in comprehensions, if the Boolean expression is **true**, we elide “| **true**”; (3) in **await** statements, if the statement in one of the clauses is **skip**, we elide “: **skip**”.

### Notation used in the grammar.

- A symbol in the grammar is a terminal symbol if it starts with a lower-case letter.
- A symbol in the grammar is a non-terminal symbol if it starts with an upper-case letter.
- | separates alternatives.
- \* after a non-terminal means “0 or more occurrences”
- + after a non-terminal means “1 or more occurrences”

### Well-formedness requirements on programs.

1. The top-level method in a program must be named **main**. It gets executed in an instance of the **Process** class when the program starts.
2. Each label used in a receive definition must be the label of some statement that appears in the same class as the receive definition.
3. Invocations of methods defined using **def** appear only in method call statements. Invocations of methods defined using **defun** appear only in method call expressions.

### Constructs whose semantics is given by translation.

1. The semantics of await statements not preceded by explicitly specified labels is given by translation. Specifically, for each await-statement whose associated label is the empty string, we generate a fresh label name  $\ell$ , associate  $\ell$  with that await statement (by replacing the empty string with  $\ell$  in that await statement), and insert  $\ell$  in every at-clause in the class containing the await-statement.

```

Program ::= Configuration ProcessClass* Method

ProcessClass ::= class ClassName extends ClassName: Method* ReceiveDef*

ReceiveDef ::=
  | receive ReceivePattern+ at LabelName+ : Statement
  | receive ReceivePattern+ : Statement

ReceivePattern ::= Pattern from InstanceVariable

Method ::=
  def MethodName(Parameter*) Statement
  | defun MethodName(Parameter*) Expr

Statement ::=
  Label InstanceVariable = Expr
  | Label InstanceVariable = new ClassName
  | Label InstanceVariable = { Pattern : Iterator* | Expr }
  | Label Statement ; Statement
  | Label if Expr: Statement else: Statement
  | Label for Iterator: Statement
  | Label while Expr: Statement
  | Label Expr.MethodName(Expr*)
  | Label send Tuple to Expr
  | Label await Expr : Stmt AnotherAwaitClause*
  | Label await Expr : Stmt AnotherAwaitClause* timeout Expr
  | Label skip

Expr ::=
  Literal
  | Parameter
  | InstanceVariable
  | Tuple
  | Expr.MethodName(Expr*)
  | UnaryOp(Expr)
  | BinaryOp(Expr, Expr)
  | isinstance(Exp, ClassName)
  | and(Expr, Expr)          // conjunction (short-circuiting)
  | or(Expr, Expr)          // disjunction (short-circuiting)
  | each Iterator | Expr
  | some Iterator | Expr

Tuple ::= (Expr*)

```

Figure 1: Abstract Syntax, Part 1

2. The boolean operators **and** and **each** can be statically eliminated as follows, so we assume in the semantics below that the program does not contain these constructs:  $e_1$  **and**  $e_2$  is replaced with  $\text{not}(\text{not}(e_1) \text{ or } \text{not}(e_2))$ , and **each**  $iter \mid e$  is replaced with  $\text{not}(\text{some } iter \mid \text{not}(e))$ .

```

UnaryOp ::=
  | not           // boolean negation
  | isTuple       // test whether a value is a tuple
  | len           // length of a tuple

BinaryOp ::=
  is             // identity-based equality, as in Python
  | plus         // sum
  | select       // select(t,i) returns the i'th component of tuple t

Pattern ::= InstanceVariable | TuplePattern

TuplePattern ::= (PatternElement*)

PatternElement ::= Constant | InstanceVariable | =InstanceVariable

Iterator ::= Pattern in Expr

AnotherAwaitClause ::= or Expr : Stmt

Configuration ::= configuration ChannelOrder ChannelReliability ...
ChannelOrder ::= fifo | unordered
ChannelReliability ::= reliable | unreliable

ClassName ::= ...
MethodName ::= ...
Parameter ::= self | ...
InstanceVariable ::= Expression.Field
Field ::= ...
Label ::= emptyString | LabelName
LabelName ::= ...
Literal ::= BooleanLiteral | IntegerLiteral | ...
BooleanLiteral ::= true | false
IntegerLiteral ::= ...

```

Figure 2: Abstract Syntax, Part 2

3. Comprehensions can be statically eliminated as follows, so we assume in the semantics below that the program does not contain comprehensions. Specifically, the comprehension  $\ell x = \{ e \mid x_1 \text{ in } e_1, \dots, x_n \text{ in } e_n \mid b \}$ , where  $\ell$  is a label and each  $x_i$  is a pattern, can be replaced with

```

ℓ x = new Set()
for x1 in e1:
  ...
  for xn in en:
    if b:
      x.add(e)

```

4. Iterators containing tuple patterns can be statically rewritten as iterators without tuple patterns, so

we assume in the semantics below that iterators do not contain tuple patterns. The rewriting is done as follows.

- Consider the universal quantification **each**  $(e_1, \dots, e_n)$  **in**  $s \mid b$ . Let  $\theta$  be the substitution that replaces  $e_i$  with **select** $(x, i)$  for each  $i$  such that  $e_i$  is a variable not prefixed with “=”. Let  $\{j_1, \dots, j_m\}$  contain the indices of the constants and the variables prefixed with “=” in  $(e_1, \dots, e_n)$ . Let  $x$  be a fresh variable. The quantification can be rewritten as **each**  $x$  **in**  $s \mid (\text{isTuple}(x)$  **and**  $\text{len}(x)$  **is**  $n$  **and** **select** $(x, j_1)$  **is**  $e_{j_1}$  **and**  $\dots$  **and** **select** $(x, j_m)$  **is**  $e_{j_m}$ ) **implies**  $b\theta$ . Informally, the semantics is that values in  $s$  that do not match the tuple pattern are ignored by the universal quantification, i.e.,  $b$  does not need to hold for those values.
- Consider the existential quantification **some**  $(e_1, \dots, e_n)$  **in**  $s \mid b$ . Let  $\theta$  be the substitution that replaces  $e_i$  with **select** $(x, i)$  for each  $i$  such that  $e_i$  is a variable not prefixed with “=”. Let  $\{j_1, \dots, j_m\}$  contain the indices of the constants and the variables prefixed with “=” in  $(e_1, \dots, e_n)$ . Let  $x$  be a fresh variable. The quantification can be rewritten as **some**  $x$  **in**  $s \mid \text{isTuple}(x)$  **and**  $\text{len}(x)$  **is**  $n$  **and** **select** $(x, j_1)$  **is**  $e_{j_1}$  **and**  $\dots$  **and** **select** $(x, j_m)$  **is**  $e_{j_m}$  **and**  $b\theta$ .
- Consider the loop **for**  $(e_1, \dots, e_n)$  **in**  $e : s$ . Let  $\{i_1, \dots, i_k\}$  contain the indices in  $(e_1, \dots, e_n)$  of variables not prefixed with “=”. Let  $\{j_1, \dots, j_m\}$  contain the indices in  $(e_1, \dots, e_n)$  of constants and variables prefixed with “=”. Let  $\theta$  be the substitution that replaces  $e_i$  with **select** $(x, i)$  for each  $i$  in  $\{i_1, \dots, i_k\}$ . Let  $x$  and  $y$  be fresh variables. Note that  $e$  may denote a set or sequence, and duplicate bindings for the tuple of variables  $(e_{i_1}, \dots, e_{i_k})$  are filtered out if  $e$  is a set but not if  $e$  is a sequence. The loop can be rewritten as

```

if isinstance(x, Set)
    // filter out duplicate bindings.  y contains processed bindings.
    y = new Set
    for x in e :
        if (isTuple(x) and len(x) is n
            and select(x, j1) is ej1 and ...
            and select(x, jm) is ejm and bθ
            and not y.contains((select(x, i1), ..., select(x, ik)))):
            y.add((select(x, i1), ..., select(x, ik)))
            sθ
        else:
            skip
    else
        for x in e :
            if (isTuple(x) and len(x) is n
                and select(x, j1) is ej1 and ...
                and select(x, jm) is ejm and bθ):
                sθ
            else:
                skip

```

#### Notes.

1. `ClassName` must include `Process`. `Process` is a pre-defined class. It should not be defined explicitly. `Process` has fields `sent` and `received`, and it has a method `start`.

2. The grammar allows receive definitions to appear in classes that do not extend **Process**, but such receive definitions are useless, so it would be reasonable to declare them illegal.
3. Sets and sequences are treated as objects, because they are mutable. **ClassName** must include **Set** and **Sequence**. These are predefined classes that should not be defined explicitly. Methods of **Set** include **add**, **del**, **contains**, **min**, **max**, and **size**. Methods of **Sequence** include **add** (which adds an element at the end of the sequence), **contains**, and **length**. We give the semantics explicitly for a few of these methods; the others are handled similarly.
4. Tuples are treated as values, not as objects, because they are immutable.
5. Object allocation and comprehension are statements, not expressions, because they have side-effects. All expressions are side-effect free.
6. The values of method parameters cannot be updated (e.g., using assignment statements). For brevity, local variables of methods are omitted from the language. Consequently, assignment is allowed only for instance variables.
7. Semantically, the **for** loop copies the contents of a (mutable) sequence or set into an (immutable) tuple before iterating over it, to ensure that changes to the sequence or set by the loop body do not affect the iteration. An implementation could use optimizations to achieve this semantics without copying when possible.
8. For brevity, among the standard arithmetic operations (+, -, \*, etc.), we include only one representative operation in the abstract syntax and semantics; others are handled similarly.
9. The semantics below does not model real-time, so timeouts in **await** statements are simply allowed to occur non-deterministically.
10. We omit the concept of “site” (process location) from the semantics, and we omit the site argument from the constructor when creating instances of process classes, because process location does not affect other aspects of the semantics.
11. We do not include **use handling\_all** explicitly in the program syntax, but we assume its effect in the semantics, since it is the default.
12. To support initialization of a process by its parent, a process can access fields of another process and invoke methods on another process before the latter process is started but not afterward. In the future, we might extend the semantics to allow remote method invocation, i.e., to allow a process to invoke methods on another process after the latter process is started.
13. We require that all messages are tuples. This is an inessential restriction; it slightly simplifies the specification of pattern matching between messages and receive patterns.
14. A process’s **sent** sequence contains pairs of the form  $(m, d)$ , where  $m$  is a message sent by the process to destination  $d$ . A process’s **receive** sequence contains pairs of the form  $(m, s)$ , where  $m$  is a message received by the process from sender  $s$ .

## 2 Semantic Domains

The semantic domains are defined in Figure 3.

$$\begin{aligned}
\textit{Bool} &= \{\text{true}, \text{false}\} \\
\textit{Int} &= \dots \\
\textit{Address} &= \dots \\
\textit{ProcessAddress} &= \dots \\
\textit{Tuple} &= \textit{Val}^* \\
\textit{Val} &= \textit{Bool} \cup \textit{Int} \cup \textit{Address} \cup \textit{Tuple} \\
\textit{Object} &= (\text{Field} \rightarrow \textit{Val}) \cup \textit{SetOfVal} \cup \textit{SeqOfVal} \\
\textit{SetOfVal} &= \text{Set}(\textit{Val}) \\
\textit{SeqOfVal} &= \textit{Val}^* \\
\textit{MsgQueue} &= (\textit{Tuple} \times \textit{ProcessAddress})^* \\
\textit{ChannelStates} &= \textit{ProcessAddress} \times \textit{ProcessAddress} \\
&\quad \rightarrow \textit{Tuple}^* \\
\textit{HeapType} &= \textit{Address} \rightarrow \text{ClassName} \\
\textit{LocalHeap} &= \textit{Address} \rightarrow \textit{Object} \\
\textit{Heap} &= \textit{ProcessAddress} \rightarrow \textit{LocalHeap} \\
\textit{State} &= (\textit{ProcessAddress} \rightarrow \text{Statement}) \\
&\quad \times \textit{HeapType} \times \textit{Heap} \times \textit{ChannelStates} \\
&\quad \times (\textit{ProcessAddress} \rightarrow \textit{MsgQueue})
\end{aligned}$$

Figure 3: Semantic Domains

**Notation:**

- $D^*$  contains sequences of values from domain  $D$ .
- $\text{Set}(D)$  contains sets of values from domain  $D$ .
- $D_1 \rightarrow D_2$  contains partial functions from  $D_1$  to  $D_2$ .  $\text{dom}(f)$  is the domain of a partial function  $f$ .

**Notes.**

- We require  $\textit{ProcessAddress} \subseteq \textit{Address}$ .
- For  $a \in \textit{ProcessAddress}$  and  $h \in \textit{Heap}$ ,  $h(a)$  is the local heap of process  $a$ . For  $a \in \textit{Address}$  and  $ht \in \textit{HeapType}$ ,  $ht(a)$  is the type of the object with address  $a$ . For convenience, we use a single (global) function for  $\textit{HeapType}$  in the semantics, even though the information in that function is distributed in the same way as the heap itself in an implementation.
- The  $\textit{MsgQueue}$  associated with a process by the last component of a state contains messages (paired with their senders) that have arrived at the process but have not yet been received. This information is needed to express the requirement that all matching messages that have arrived at the process must be handled when execution of the process reaches a yield point.

```

Expression ::=
| Address
| Address.Field

Statement ::=
| for Variable intuple Tuple: Statement

```

Figure 4: Extensions to the abstract syntax

```

C ::=
| []
| [Val*,C,Expr*]
| Expr.MethodName(Val*,C,Expr*)
| UnaryOp(C)
| BinaryOp(C,Expr)
| BinaryOp(Val,C)
| instanceof(C,ClassName)
| or(C,Expr)
| some Pattern in C | Expr
| C.Field = Expression
| Address.Field = C
| InstanceVariable = C
| C ; Statement
| if C: Statement else: Statement
| for InstanceVariable in C: Statement
| for InstanceVariable intuple Tuple: C
| send C to Expr
| send Val to C
| await Expr : Stmt AnotherAwaitClause* timeout C

```

Figure 5: Evaluation contexts

### 3 Extended Abstract Syntax

Section 1 defines the abstract syntax of programs that can be written by the user. Figure 4 extends the abstract syntax to include additional forms into which programs may evolve during evaluation. Only the new productions are shown here; all of the productions given above carry over unchanged.

The statement `for  $v$  intuple  $t$ :  $s$`  iterates over the elements of tuple  $t$ , in the obvious way.

### 4 Evaluation Contexts

Evaluation contexts, also called reduction contexts, are used to identify the next part of an expression or statement to be evaluated. An evaluation context is an expression or statement with a hole, denoted `[]`, in place of the next sub-expression or sub-statement to be evaluated. Evaluation contexts are defined in Figure 5.

## 5 Transition Relations

The transition relation for expressions has the form  $e \rightarrow_{ht,h} e'$ , where  $e$  and  $e'$  are expressions,  $ht \in \text{HeapType}$ , and  $h \in \text{LocalHeap}$ .

The transition relation for statements has the form  $\sigma \rightarrow \sigma'$  where  $\sigma \in \text{State}$  and  $\sigma' \in \text{State}$ . Both transition relations are implicitly parameterized by the program, which is needed to look up method definitions and configuration information. The transition relation for expressions is defined in Figure 6. The transition relation for statements is defined in Figures 7–9.

### Notation and auxiliary functions.

- In the transition rules,  $a$  matches an address;  $p$  matches a primitive value (i.e., a boolean value, integer value, or an address);  $v$  matches a value (i.e., a primitive value or a tuple whose components are values); and  $\ell$  matches a label.
- For an expression or statement  $e$ ,  $e[x := y]$  denotes  $e$  with all occurrences of  $x$  replaced with  $y$ .
- A function matches the pattern  $f[x \rightarrow y]$  if  $f(x)$  equals  $y$ .
- $f[x := y]$  denotes the function that is the same as  $f$  except that it maps  $x$  to  $y$ .
- $f_0$  denotes the empty partial function, i.e., the partial function whose domain is the empty set.
- For a (partial) function  $f$ ,  $f \ominus a$  denotes the function that is the same as  $f$  except that it has no mapping for  $a$ .
- Sequences are denoted with angle brackets, e.g.,  $\langle 0, 1, 2 \rangle \in \text{Int}^*$ .
- $s@t$  is the concatenation of sequences  $s$  and  $t$ .
- $\text{tail}(s)$  is the tail of sequence  $s$ , i.e., the sequence obtained by removing the first element of  $s$ .
- $\text{first}(s)$  is the first element of sequence  $s$ .
- $\text{length}(s)$  is the length of sequence  $s$ .
- $\text{extends}(c_1, c_2)$  holds iff class  $c_1$  is a descendant of class  $c_2$  in the inheritance hierarchy.
- For  $c \in \text{ClassName}$ ,

$$\text{new}(c) = \begin{cases} \{\} & \text{if } c = \text{Set} \\ \langle \rangle & \text{if } c = \text{Sequence} \\ f_0 & \text{otherwise} \end{cases}$$

- For  $m \in \text{MethodName}$  and  $c \in \text{ClassName}$ ,  $\text{methodDef}(c, m, \text{def})$  holds if (1) class  $c$  defines method  $m$ , and  $\text{def}$  is the definition of  $m$  in  $c$ , or (2)  $c$  does not define  $m$ , and  $\text{def}$  is the definition of  $m$  in the nearest ancestor of  $c$  in the inheritance hierarchy that defines  $m$ .
- For  $h, \bar{h}, \bar{h}' \in \text{LocalHeap}$  and  $ht, ht' \in \text{HeapType}$  and  $v, v' \in \text{Val}$ ,  $\text{isCopy}(v, h, \bar{h}, ht, v', \bar{h}', ht')$  holds if (1)  $v$  is a value for a process with local heap  $h$  (i.e., addresses in  $v$  are evaluated with respect to  $h$ ), (2)  $v'$  is a copy of  $v$  for a process with local heap  $\bar{h}'$ , i.e.,  $v'$  is the same as  $v$  except that, instead of referencing objects in  $h$ , it references newly allocated copies of those objects in  $\bar{h}'$ , and (3)  $\bar{h}'$  and  $ht'$  are versions of  $\bar{h}$  and  $ht$  updated to reflect the allocation of those objects. As an exception, because process addresses are used as global identifiers, process addresses in  $v$  are copied unchanged into  $v'$ ,



and new copies of process objects are not created. We give some auxiliary definitions and then a formal definition of *isCopy*. For  $v \in \text{Val}$ , let  $\text{addrs}(v, h)$  denote the set of addresses that appear in  $v$  or in any objects or values reachable from  $v$  with respect to local heap  $h$ ; formally,

$$\begin{aligned} a \in \text{addrs}(v, h) &= v \in \text{Address} \wedge v = a \\ &\vee v \in \text{dom}(h) \wedge h(v) \in \text{Field} \rightarrow \text{Val} \wedge (\exists f \in \text{dom}(h(v)). a \in \text{addrs}(h(v)(f), h)) \\ &\vee v \in \text{dom}(h) \wedge h(v) \in (\text{SetOfVal} \cup \text{SeqOfVal}) \wedge (\exists v' \in h(v). a \in \text{addrs}(v', h)) \\ &\vee (\exists v_1, \dots, v_n \in \text{Val}. \exists i \in [1..n]. v = (v_1, \dots, v_n) \wedge a \in \text{addrs}(v_i, h)) \end{aligned}$$

For  $v, v' \in \text{Val}$  and  $f \in \text{Address} \rightarrow \text{Address}$ ,  $\text{subst}(v, v', f)$  holds if  $v$  is obtained from  $v'$  by replacing each occurrence of an address  $a$  in  $\text{dom}(f)$  with  $f(a)$  (informally,  $f$  maps addresses of new objects in  $v'$  to addresses of corresponding old objects in  $v$ ); formally,

$$\begin{aligned} \text{subst}(v, v', f) &= v \in \text{Bool} \cup \text{Int} \cup (\text{Address} \setminus \text{dom}(f) \wedge v' = v) \\ &\vee v \in \text{dom}(f) \wedge f(v') = v \\ &\vee (\exists v_1, \dots, v_n, v'_1, \dots, v'_n. v = (v_1, \dots, v_n) \wedge v' = (v'_1, \dots, v'_n) \wedge (\forall i \in 1..n. \text{subst}(v_i, v'_i, f))) \end{aligned}$$

Similarly, for  $o, o' \in \text{Object}$  and  $f \in \text{Address} \rightarrow \text{Address}$ ,  $\text{subst}(o, o', f)$  holds if  $o$  is obtained from  $o'$  by replacing each occurrence of an address  $a$  in  $\text{dom}(f)$  with  $f(a)$ . For sets  $S$  and  $S'$ , let  $S \xrightarrow{1-1} S'$  be the set of bijections between  $S$  and  $S'$ . Finally, *isCopy* is defined as follows (informally,  $A$  contains the addresses of the newly allocated objects):

$$\begin{aligned} \text{isCopy}(v, h, \bar{h}, ht, v', \bar{h}', ht') &= \exists A \subset \text{Address} \setminus \text{ProcessAddress}, f \in A \xrightarrow{1-1} (\text{addrs}(v, h) \setminus \text{ProcessAddress}). \\ &A \cap \text{dom}(ht) = \emptyset \wedge \text{dom}(ht') = \text{dom}(ht) \cup A \wedge \text{dom}(\bar{h}') = \text{dom}(\bar{h}) \cup A \\ &\wedge (\forall a \in \text{dom}(ht). ht'(a) = ht(a)) \\ &\wedge (\forall a \in \text{dom}(\bar{h}). \bar{h}'(a) = \bar{h}(a)) \\ &\wedge (\forall a \in A. ht'(a) = ht(f(a)) \wedge \text{subst}(\bar{h}(a), \bar{h}'(a), f)) \end{aligned}$$

- For  $m \in \text{Val}$ ,  $a \in \text{ProcessAddress}$ ,  $\ell \in \text{LabelName}$ ,  $h \in \text{LocalHeap}$ , and a receive definition  $d$ , if message  $m$  can be received from  $a$  at label  $\ell$  by a process with local heap  $h$  using receive definition  $d$ , then  $\text{matchRcvDef}(m, a, \ell, h, d)$  returns the appropriately instantiated body of  $d$ . Specifically, if (1) either  $d$  lacks an **at** clause, or  $d$  has an **at** clause that includes  $\ell$  (note: this implies that  $\ell$  is a label name and is not the empty string), and (2)  $d$  contains a receive pattern  $P$  **from**  $x$  such that there exists a substitution  $\theta$  such that (2a)  $m = P\theta$  and (2b)  $\theta(y) = h(y)$  for every variable  $y$  prefixed with “=” in  $P$ , then, letting  $\theta$  be the substitution obtained using the first receive pattern in  $d$  for which (2) holds,  $\text{matchRcvDef}(m, a, \ell, h, d)$  returns  $s\theta[x := a]$ , where  $s$  is the body of  $d$  (i.e., the statement that appears in  $d$ ). Otherwise,  $\text{matchRcvDef}(m, a, \ell, d)$  returns  $\perp$ .
- For  $m, \bar{m} \in \text{Val}$ ,  $a \in \text{ProcessAddress}$ ,  $\ell \in \text{LabelName}$ ,  $c \in \text{ClassName}$ , and  $h \in \text{LocalHeap}$ , if message  $m$  can be received from  $a$  at label  $\ell$  in class  $c$  by a process with local heap  $h$ , and  $\bar{m}$  is a copy of  $m$ , then  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$  returns a statement that should be executed when receiving  $m$  in that context. Specifically, if class  $c$  contains a receive definition  $d$  such that  $\text{matchRcvDef}(m, a, \ell, h, d)$  is not  $\perp$ , then, letting  $d_1, \dots, d_n$  be the sequence of receive definitions  $d$  in  $c$  (in the order that they appear in  $c$ ) such that  $\text{matchRcvDef}(m, a, \ell, h, d)$  is not  $\perp$ , and letting  $s_i = \text{matchRcvDef}(m, \bar{m}, a, \ell, h, d_i)$ ,  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$  returns **self.received.add**(( $\bar{m}, a$ ));  $s_1$ ;  $\dots$ ;  $s_n$ . Otherwise, if  $\ell$  is not the empty string, and there does not exist a receive definition  $d$  in  $c$  and a label  $\ell'$  such that  $\text{matchRcvDef}(m, a, \ell', h, d)$  holds, then  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$  returns **self.received.add**(( $\bar{m}, a$ )). Otherwise,  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$  returns  $\perp$ .

- For  $q \in \text{MsgQueue}$ ,  $\ell \in \text{LabelName}$ ,  $c \in \text{ClassName}$ , and  $h \in \text{LocalHeap}$ , if  $q$  contains an entry  $(m, \bar{m}, a)$  such that  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$  is not  $\perp$ , then, letting  $(m, \bar{m}, a)$  be the first such entry in  $q$ ,  $\text{rcvMsg}(q, \ell, c, h)$  returns  $(q', s)$ , where  $q'$  is obtained from  $q$  by removing  $(m, \bar{m}, a)$ , and  $s$  is the statement returned by  $\text{rcvAtLabel}(m, \bar{m}, a, \ell, c, h)$ . Otherwise,  $\text{rcvMsg}(q, \ell, c, h)$  returns  $\perp$ .

## 6 Executions

An execution is a sequence of transitions  $\sigma_0 \rightarrow \sigma_1 \rightarrow \sigma_2 \rightarrow \dots$  such that  $\sigma_0$  is an initial state. The set of initial states is defined in Figure 10. Intuitively,  $a_p$  is the address of the initial process,  $a_r$  is the address of the **received** sequence of the initial process, and  $a_s$  is the address of the **sent** sequence of the initial process.

Informally, execution of the statement initially associated with a process may eventually (1) terminate (i.e., the statement associated with the process becomes **skip**, indicating that there is nothing left for the process to do), (2) get stuck (i.e., the statement associated with the process is not **skip**, and the process has no enabled transitions) due to an unsatisfied **await** statement or an error (e.g., the statement contains an expression that tries to select a component from a value that is not a tuple) or (3) run forever due to an infinite loop or infinite recursion.

## Acknowledgments

The semantics is based on discussions with Yanhong Liu, DistAlgo's main designer. Bo Lin and Xuetian Weng provided helpful suggestions.

## References

- [LSLG12] Yanhong A. Liu, Scott D. Stoller, Bo Lin, and Michael Gorbavitski. From clarity to efficiency for distributed algorithms. In *Proceedings of the 2012 ACM International Conference on Object Oriented Programming Systems Languages and Applications (OOPSLA)*, pages 395–410. ACM Press, October 2012.
- [SRM09] Traian Florin Șerbănuță, Grigore Roșu, and José Meseguer. A rewriting logic approach to operational semantics. *Information and Computation*, 207:305–340, 2009.
- [WF94] Andrew K. Wright and Matthias Felleisen. A syntactic approach to type soundness. *Information and Computation*, 115:38–94, 1994.

```

% primitive value
 $p \rightarrow_{ht,h} p$ 

% field access
 $a.f \rightarrow_{ht,h} h(a)(f) \quad \text{if } a \in \text{dom}(h) \wedge f \in \text{dom}(h(a))$ 

% tuple
 $(v_1, \dots, v_n) \rightarrow_{ht,h} (v_1, \dots, v_n)$ 

% invoke method in user-defined class
 $a.m(v_1, \dots, v_n) \rightarrow_{ht,h} e[\mathbf{self} := a, x_1 := v_1, \dots, x_n := v_n]$ 
if  $\exists c. ht(a) = c \wedge \text{methodDef}(c, m, \mathbf{defun} \ m(x_1, \dots, x_n) \ e)$ 

% invoke method in pre-defined class (representative examples)
 $a.\mathbf{contains}(v_1) \rightarrow_{ht,h} \mathbf{true} \quad \text{if } \exists S. ht(a) = \mathbf{Set} \wedge h(a) = S \wedge v_1 \in S$ 
 $a.\mathbf{contains}(v_1) \rightarrow_{ht,h} \mathbf{false} \quad \text{if } \exists S. ht(a) = \mathbf{Set} \wedge h(a) = S \wedge v_1 \notin S$ 

% unary operations
 $\mathbf{not}(\mathbf{true}) \rightarrow_{ht,h} \mathbf{false}$ 
 $\mathbf{not}(\mathbf{false}) \rightarrow_{ht,h} \mathbf{true}$ 
 $\mathbf{isTuple}(v) \rightarrow_{ht,h} \mathbf{true} \quad \text{if } v \text{ is a tuple}$ 
 $\mathbf{isTuple}(v) \rightarrow_{ht,h} \mathbf{false} \quad \text{if } v \text{ is not a tuple}$ 
 $\mathbf{len}(v) \rightarrow_{ht,h} n \quad \text{if } v \text{ is a tuple with } n \text{ components}$ 

% binary operations
 $\mathbf{is}(v_1, v_2) \rightarrow_{ht,h} \mathbf{true} \quad \text{if } v_1 \text{ and } v_2 \text{ are the same value}$ 

 $\mathbf{plus}(v_1, v_2) \rightarrow_{ht,h} v_3 \quad \text{if } v_1 \in \text{Int} \wedge v_2 \in \text{Int} \wedge v_3 = v_1 + v_2$ 

 $\mathbf{select}(v_1, v_2) \rightarrow_{ht,h} v_3$ 
if  $v_2 \in \text{Int} \wedge v_2 > 0 \wedge (v_1 \text{ is a tuple with at least } v_2 \text{ components}) \wedge (v_3 \text{ is the } v_2\text{'th component of } v_1)$ 

% instanceof
 $\mathbf{isinstance}(a, c) \rightarrow_{ht,h} \mathbf{true} \quad \text{if } ht(a) = c$ 
 $\mathbf{isinstance}(a, c) \rightarrow_{ht,h} \mathbf{false} \quad \text{if } ht(a) \neq c$ 

% disjunction
 $\mathbf{or}(\mathbf{true}, e) \rightarrow_{ht,h} \mathbf{true}$ 
 $\mathbf{or}(\mathbf{false}, e) \rightarrow_{ht,h} e$ 

% existential quantification
 $\mathbf{some} \ x \ \mathbf{in} \ a \mid e \rightarrow_{ht,h} e[x := v_1] \ \mathbf{or} \ \dots \ \mathbf{or} \ e[x := v_n]$ 
if  $(ht(a) = \mathbf{Sequence} \wedge h(a) = \langle v_1, \dots, v_n \rangle) \vee (ht(a) = \mathbf{Set} \wedge \langle v_1, \dots, v_n \rangle \text{ is a linearization of } h(a))$ 

```

Figure 6: Transition relation for expressions.

```

% field assignment
( $P[a \rightarrow \ell \ a'.f = v], ht, h[a \rightarrow ha[a' \rightarrow o]], ch, mq$ )  $\rightarrow$  ( $P[a := \mathbf{skip}], ht, h[a := ha[a' := o[f := v]]], ch, mq$ )
if  $rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% object allocation
( $P[a \rightarrow \ell \ a'.f = \mathbf{new} \ c], ht, h[a \rightarrow ha[a' \rightarrow o]], ch, mq$ )
 $\rightarrow$  ( $P[a := \mathbf{skip}], ht[a' := c], h[a := ha[a' := o[f := a_c], a_c := \mathbf{new}(c)]], ch, mq$ )
if  $a_c \notin \text{dom}(ht) \wedge a_c \in \text{Address} \wedge (a_c \in \text{ProcessAddress} \iff \text{extends}(c, \text{Process}))$ 
 $\wedge rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% sequential composition
( $P[a \rightarrow \ell \ \mathbf{skip}; s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s], ht, h, ch, mq$ )
if  $rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% conditional statement
( $P[a \rightarrow \ell \ \mathbf{if} \ \mathbf{true} : s_1 \ \mathbf{else} : s_2], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s_1], ht, h, ch, mq$ )
if  $rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

( $P[a \rightarrow \ell \ \mathbf{if} \ \mathbf{false} : s_1 \ \mathbf{else} : s_2], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s_2], ht, h, ch, mq$ )
if  $rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% for loop
( $P[a \rightarrow \ell \ \mathbf{for} \ x \ \mathbf{in} \ a' : s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := \mathbf{for} \ x \ \mathbf{intuple} \ (v_1, \dots, v_n) : s], ht, h, ch, mq$ )
if  $((ht(a) = \mathbf{Sequence} \wedge h(a)(a') = \langle v_1, \dots, v_n \rangle) \vee (ht(a) = \mathbf{Set} \wedge \langle v_1, \dots, v_n \rangle \text{ is a linearization of } h(a)(a')))$ 
 $\wedge rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

( $P[a \rightarrow \mathbf{for} \ x \ \mathbf{intuple} \ (v_1, \dots, v_n) : s], ht, h, ch, mq$ )
 $\rightarrow$  ( $P[a := s[x := v_1]; \mathbf{for} \ x \ \mathbf{intuple} \ (v_2, \dots, v_n) : s], ht, h, ch, mq$ )

( $P[a \rightarrow \mathbf{for} \ x \ \mathbf{intuple} \ () : s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := \mathbf{skip}], ht, h, ch, mq$ )

% while loop
( $P[a \rightarrow \ell \ \mathbf{while} \ e : s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := \mathbf{if} \ e : (s; \mathbf{while} \ e : s) \ \mathbf{else} : \mathbf{skip}], ht, h, ch, mq$ )
if  $rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% invoke method in user-defined class
( $P[a \rightarrow \ell \ a'.m(v_1, \dots, v_n)], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s[\mathbf{self} := a, x_1 := v_1, \dots, x_n := v_n]], ht, h, ch, mq$ )
if  $a' \in \text{dom}(h(a))$ 
 $\wedge ht(a') \notin \{\mathbf{Process}, \mathbf{Set}, \mathbf{Sequence}\} \wedge \text{methodDef}(ht(a'), m, \mathbf{def} \ m(x_1, \dots, x_n) \ s)$ 
 $\wedge rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

% invoke method in pre-defined class (representative examples)

% Process.start allocates a local heap and sent and received sequences for the new process,
% and moves the started process to the new local heap
( $P[a \rightarrow \ell \ a'.\mathbf{start}()], ht, h[a \rightarrow ha[a' \rightarrow o]], ch, mq$ )
 $\rightarrow$  ( $P[a := \mathbf{skip}, a' := a'.\mathbf{run}()], ht[a_s := \mathbf{Sequence}, a_r := \mathbf{Sequence}],$ 
 $h[a := ha \ominus a', a' := f_0[a' \rightarrow o][\mathbf{sent} := a_s, \mathbf{received} := a_r], a_r := \langle \rangle, a_s := \langle \rangle], ch, mq$ )
if  $\text{extends}(ht(a'), \mathbf{Process}) \wedge (ht(a') \text{ inherits } \mathbf{start} \text{ from } \mathbf{Process}) \wedge a_r \notin \text{dom}(ht) \wedge a_s \notin \text{dom}(ht)$ 
 $\wedge a_r \in \text{Address} \setminus \text{ProcessAddress} \wedge a_s \in \text{Address} \setminus \text{ProcessAddress} \wedge rcvMsg(mq(a), \ell, ht(a), h(a)) = \perp$ 

```

Figure 7: Transition relation for statements, part 1.

```

% invoke method in pre-defined class (representative examples, continued)
( $P[a \rightarrow \ell \text{ a'}.\text{add}(v_1)], ht, h[a \rightarrow ha], ch, mq$ )  $\rightarrow$  ( $P[a := \text{skip}], ht, h[a := ha[a' := ha(a') \cup \{v_1\}]], ch, mq$ )
if  $a' \in \text{dom}(ha) \wedge ht(a') = \text{Set} \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

( $P[a \rightarrow \ell \text{ a'}.\text{append}(v_1)], ht, h[a \rightarrow ha], ch, mq$ )  $\rightarrow$  ( $P[a := \text{skip}], ht, h[a := ha[a' := ha(a')@ha(v_1)]], ch, mq$ )
if  $a' \in \text{dom}(ha) \wedge ht(a') = \text{Sequence} \wedge ht(v_1) = \text{Sequence} \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

% send to a process. create copies of the message for the sender's send history and the receiver.
( $P[a \rightarrow \ell \text{ send } v \text{ to } a'], ht, h[a \rightarrow ha, a' \rightarrow ha'], ch, mq$ )
 $\rightarrow$  ( $P[a := \text{skip}], ht_2, h[a := ha_1[a_s := ha(a_s)@((v_1, a'))]], a' := ha'_1], ch[(a, a') := (v')@ch((a, a'))], mq$ )
if  $a' \in \text{ProcessAddress} \wedge a_s = ha(a)(\text{sent}) \wedge \text{isCopy}(v, ha, ha, ht, v_1, ha_1, ht_1)$ 
 $\wedge \text{isCopy}(v, ha_1, ha', ht_1, v', ha'_1, ht_2) \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

% send to a set of processes
( $P[a \rightarrow \ell \text{ send } v \text{ to } a'], ht, h[a \rightarrow ha], ch, mq$ )
 $\rightarrow$  ( $P[a := \text{for } x \text{ in } a': \text{send } v \text{ to } x], ht, h[a := ha[a_s := ha(a_s)@((v, a'))]], ch, mq$ )
if  $ht(a') = \text{Set} \wedge a_s = ha(a)(\text{sent}) \wedge (x \text{ is a fresh variable}) \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

% message reordering
( $P, ht, h, ch[(a, a') \rightarrow q], mq$ )  $\rightarrow$  ( $P, ht, h, ch[(a, a') := q'], mq$ )
if (channel order is unordered in the program configuration)  $\wedge$  ( $q'$  is a permutation of  $q$ )

% message loss
( $P, ht, h, ch[(a, a') \rightarrow q], mq$ )  $\rightarrow$  ( $P, ht, h, ch[(a, a') := q'], mq$ )
if (channel reliability is unreliable in the program configuration)  $\wedge$  ( $q'$  is a subsequence of  $q$ )

% receive a message
( $P[a \rightarrow \ell s], ht, h, ch, mq[a \rightarrow q]$ )  $\rightarrow$  ( $P[a := s'[\text{self} := a]; \ell s], ht, h, ch, mq[a := q']$ )
if  $\text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = (q', s')$ 

% arrival of message at process (message moves from channel to message queue)
% append tuple  $(m, m', a)$  to message queue.  $m'$  is a copy of message  $m$  for storage in receive history.
( $P, ht, h[a' \rightarrow ha], ch[(a, a') \rightarrow q], mq$ )
 $\rightarrow$  ( $P, ht', h[a' \rightarrow ha'], ch[(a, a') := \text{tail}(q)], mq[a' := mq(a')@((\text{first}(q), m', a))]$ )
if  $\text{length}(q) > 0 \wedge \text{isCopy}(\text{first}(q), ha, ha, ht, m', ha', ht')$ 

% await without timeout clause
( $P[a \rightarrow \ell \text{ await } e_1:s_1 \text{ or } \dots \text{ or } e_n:s_n], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s_i], ht, h, ch, mq$ )
if  $i \in [1..n] \wedge e_i \rightarrow_{ht, h(a)} \text{true} \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

% await with timeout clause, terminated by true condition
( $P[a \rightarrow \ell \text{ await } e_1:s_1 \text{ or } \dots \text{ or } e_n:s_n \text{ timeout } v:s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s_i], ht, h, ch, mq$ )
if  $i \in [1..n] \wedge e_i \rightarrow_{ht, h(a)} \text{true} \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

% await with timeout clause, terminated by timeout (occurs non-deterministically)
( $P[a \rightarrow \ell \text{ await } e_1:s_1 \text{ or } \dots \text{ or } e_n:s_n \text{ timeout } v:s], ht, h, ch, mq$ )  $\rightarrow$  ( $P[a := s], ht, h, ch, mq$ )
if  $e_1 \rightarrow_{ht, h(a)} \text{false} \wedge \dots \wedge e_n \rightarrow_{ht, h(a)} \text{false} \wedge \text{rcvMsg}(mq(a), \ell, ht(a), h(a)) = \perp$ 

```

Figure 8: Transition relation for statements, part 2.

$$\begin{array}{c}
\% \text{ context rule for expressions} \\
\frac{e \rightarrow_{ht, h(a)} e'}{(P[a \rightarrow C[e]], ht, h, ch, mq) \rightarrow (P[a := C[e']], ht, h, ch, mq)} \\
\\
\% \text{ context rule for statements} \\
\frac{(P[a \rightarrow s], ht, h, ch, mq) \rightarrow (P[a := s'], ht', h', ch', mq')}{(P[a \rightarrow C[s]], ht, h, ch, mq) \rightarrow (P[a := C[s']], ht', h', ch', mq')}
\end{array}$$

Figure 9: Transition relation for statements, part 3.

$$\begin{aligned}
Init = & \\
& \{(P, ht, h, ch, mq) \in State \mid \\
& \exists a_p \in ProcessAddress, \\
& a_r \in Address \setminus ProcessAddress, \\
& a_s \in Address \setminus ProcessAddress. \\
& a_r \neq a_s \wedge \\
& P = f_0[a_p := a_p.\mathbf{main}()] \wedge \\
& ht = f_0[a_p := \mathbf{Process}, a_r := \mathbf{Sequence}, a_s := \mathbf{Sequence}] \wedge \\
& h = f_0[a_p := ha] \wedge \\
& ch = (\lambda(a_1, a_2) \in ProcessAddress \times ProcessAddress. \langle \rangle) \wedge \\
& mq = (\lambda a \in ProcessAddress. \langle \rangle) \\
& \text{where } ha = f_0[a_p := o_p, a_r := \langle \rangle, a_s := \langle \rangle] \\
& o_p = f_0[\mathbf{received} := a_r, \mathbf{sent} := a_s]\}
\end{aligned}$$

Figure 10: Initial states.